Perusharjoitus  # Visual Basic -kieleen ja kääntäjään tutustuminen
GIS-luokan koneille on asennettu Micrososft Visual Studio V6.0, joka sisältää Visual Basic, C/C++ ja Java kääntäjät / kehitysympäristöt.

Perusharjoituksissa käytetään ohjelmia, jotka on kirjoitettu Visual Basic-kielellä. Joitakin laskennallisesti vaativia ohjelmien toimintoja on toteutettu C/C++-kielisillä ohjelmilla, jotka on käännetty windows-järjestelmän dll-ohjelmakirjastoiksi (dynamic link library).

Visual Basic on tulkattava kieli, jonka takia sen suoristusajat ovat pitkiä, mutta toisaalta ohjelmankehitys vaivattomampaa, koska tulkin alla ohjelman suoritus etenee rivi kerrallaan ja virhetilanteen sattuessa virheen aiheuttaneen ohjelmarivin voi ohittaa ja/tai korjata. Visual Basic -ohjelman voi kääntää myös ns. executable-muotoon (*.exe), joskin kyse ei ole aidosta, itsenäisestä konekielisestä ohjelmasta, vaan käyttöjärjestelmä vaatii ohjelman suoristusta varten lisämoduuleja. 

Harjoituksessa tehdään esimerkkiohjelma, jonka avulla tutustutaan VB:n tietotyyppeihin, tiedosto I/O-toimintoihin, aliohjelma- ja funktiokutsuihin sekä joihinkin ohjausrakenteisiin.
Avaa VB-ympäristö

Start | Programs | Microsoft Visual Studio 6.0 | Microsoft Visual Basic 6.0 
Valitse avautuvalta välilehdeltä "New" → "Standard EXE" → "Open"
VB-ympäristö loi valmiin pohjan Win32-sovellukselle, jossa on ohjelmaikkuna nimeltään Form1.

Aktivoi Form1-ikkuna klikkaamalla sitä ja paina F4 (View | Properties window). Saat esille tätä oliota koskevien ominaisuuksien listan. Vaihda kohtaan "Caption" tekstin "Form1" tilalle "Ensimmäinen sovellukseni". Teksti Form1-ohjelmaikkunan yläpalkissa vaihtui. 

[image: image1.png]E|

Gerera





Lisätään seuraavaksi Form1-ikkunaan komentopainike: Valitse hiiren avulla näytöllä "kelluvasta" ToolBox-ikkunasta komentopainike olio ja "maalaa" se hiirellä Form1-ikkunaan. Sen "Caption" -teksti on oletusarvoisesti "Command1" ja olion nimi Form1.Command1. 
Jos kaksoisklikkaat komentopainiketta saat auki ohjelmakoodi-ikkunan, johon seuraavaksi kirjoitetaan basic-koodia tapahtumaan "Click" eli koodi joka suoritetaan, kun painiketta painetaan hiirellä ohjelmaa suoritettaessa.

Private Sub Command1_Click()

MsgBox ("Hello world")

End Sub
Nyt tämä ohjelma voidaan suorittaa: Anna VB-kehitysympäristön valikkokomento Run | Start (tai "F5" -pikanäppäinyhdistelmä).
(Tässä vaiheessa VB saattaa pyytää tallentamaan projektin, joka koostuu osista form1.frm ja project1.vbp)

Klikattaessa Command1-komentopainiketta saat msgbox-viestin "Hello World".

Ohjelman suoritus saadaan loppumaan VB-valikkokomennolla (Run | End).

Tutustutaan seuraavaksi käsitteisiin "tapahtuma" ja olion "ominaisuuksien muuttaminen". Kaksoisklikkaa Command1-painiketta saadaksesi ohjelmakoodi-ikkunan auki. Etsi oikeanpuoleisesta alasvetovalikosta tapahtuma "MouseMove", jolloin saadaan esille aliohjelma Private Sub Command1_MouseMove(), johon voidaan kirjoittaa koodia, joka suoritetaan, kun käyttöjärjestelmälle välittyy tapahtuma, jossa hiiri viedään komentopainikkeen päälle. Lisää ao. kaksi riviä ja kokeile ohjelmaa (F5).
Private Sub Command1_MouseMove(Button As Integer, Shift As Integer, X As Single, Y As Single)

Command1.Top = Command1.Top + (Rnd() - 0.5) * 100

Command1.Left = Command1.Left + (Rnd() - 0.5) * 100

End Sub
Poista lopuksi tämä ohjelmakoodi tapahtumasta MouseMove.
( Windows on tapahtumaorientoitunut käyttöjärjestelmä, joka "pyörii ikuisessa luupissa odottaen tapahtumia, jotka aikaansaavat suoritettavaa koodia". Visual Basic ohjelmassa ei ole "pääohjelmaa" tai Main()-rutiinia. Sellainen on tosin mahdollista luoda ja tehdä VB-sovellus, joka käynnistyttyään alkaa suorittaa koodia ilman käyttäjän luomaa tapahtumaa. 
Tietotyypit

VB tukee tavallisimpia tietotyyppejä: 

- etumerkitön kokonaisluku: 8 bitin "Byte"

- etumerkilliset kokonaisluvut: 16 bitin "Integer" ja 32 bitin "Long"

- liukulukutyypit: 32 bitin "single" ja 64 bitin "double"

- määrämittaiset merkkijonot: "string*#", jossa # = merkkien määrä

- vapaamittaiset merkkijonot: "string"

Lisäksi on joukko erikoistyyppejä "Currency", "Decimal", "Date" ja "Object"

C-kielen tapaan (Struct) käyttäjä voi konstruoida näistä standardityypeista omia kokonaisuuksiaan "Type".

Kaikista tietotyypeista voi tehdä taulukoita (Arrays), joilla on maksimikokona noin 270 Mtavua (vrt. ilmakuvien koko!). Taulukoilla on ominaisuuksia, jotka helpottavat ohjelmointi. Esim. taulukon indeksien ala- ja ylärajan arvoa voi käyttää muuttujana (eli sitä ei tarvitse muistaa). Taulukon alkioille voi varata tilaa lisää suorituksen aikana.
Osoitin-muuttuja VB:stä puuttuu. 

Tutustuminen tietotyyppeihin ja tiedosto I/O toimintoihin
Lisää Form1-ikkunaan toinen komentopainike "Command2".

Lisää projektiin BAS-moduuli, johon voi kirjoittaa yleisiä aliohjelmia ja funktioita sekä alustaa globaaleja (kaikkialle näkyviä) muuttujia, omia tietotyyppejä ja antaa vakioita. VB-valikkokomento (Project | Add Module).

Lisää Module1.Bas koodi-ikkunaan "General"/"Declarations", seuraavat esittelyt:
Public EtumerkitonTavu As Byte

Public EtumerkillKokLuku As Integer

Public EtumerkillIsoKokLuku As Long

Public PieniLiukuLuku As Single

Public Isoliukuluku As Double

Public MaaramittMerkkijono As String * 8

Public VapaaMerkkijono As String

Type OmaTyyppi

 No As Long

 Type As Byte

 x As Double

 y As Double

 z As Single

End Type

Public PisteTaulukko(1 To 100) As OmaTyyppi

"Public" määrittely tekee ko. muuttujista globaaleja, eli niiden senhetkinen arvo on saavutettavissa ja muutettavissa mistä ohjelman osasta (aliohjelma tai funktio) tahansa.

Kirjoita Command2-painikkeelle "Click" tapahtumaan seuraava koodi:
Private Sub Command2_Click()

Rem Tietotyypit; esitelty MODULE1.BAS :ssa; globaaleja muuttujia

EtumerkitonTavu = 1

EtumerkillKokLuku = 1

EtumerkillIsoKokLuku = 1

PieniLiukuLuku = 1.00000000000001

Isoliukuluku = 1.00000000000001

MaaramittMerkkijono = "12345678"

VapaaMerkkijono = MaaramittMerkkijono & " ovat lukuja"

Rem Avataan tiedostoja
Rem Ensin ASCII / tekstitiedostoja eri kanaville / tietovirtoihin

Open "C:\TEMP\Etumerkitonluku.txt" For Output As 1

Open "C:\TEMP\EtumerkillKokLuku.txt" For Output As 2

Open "C:\TEMP\EtumerkillIsoKokLuku.txt" For Output As 3

Open "C:\TEMP\PieniLiukuLuku.txt" For Output As 4

Open "C:\TEMP\IsoLiukuLuku.txt" For Output As 5

Open "C:\TEMP\MaaramittMerkkijono.txt" For Output As 6

Open "C:\TEMP\VapaaMerkkijono.txt" For Output As 7

Rem Kirjoitetaan muuttujien sisältö ja suljetaan tiedostot

Print #1, EtumerkitonTavu

Print #2, EtumerkillKokLuku

Print #3, EtumerkillIsoKokLuku

Print #4, PieniLiukuLuku

Print #5, Isoliukuluku

Print #6, MaaramittMerkkijono

Print #7, VapaaMerkkijono

Close 1, 2, 3, 4, 5, 6, 7

Rem Avataan Binääritiedostoja eri kanaville / tietovirtoihin

Open "C:\TEMP\Etumerkitonluku.bin" For Binary As 1

Open "C:\TEMP\EtumerkillKokLuku.bin" For Binary As 2

Open "C:\TEMP\EtumerkillIsoKokLuku.bin" For Binary As 3

Open "C:\TEMP\PieniLiukuLuku.bin" For Binary As 4

Open "C:\TEMP\IsoLiukuLuku.bin" For Binary As 5

Open "C:\TEMP\MaaramittMerkkijono.bin" For Binary As 6

Open "C:\TEMP\VapaaMerkkijono.bin" For Binary As 7

Put #1, , EtumerkitonTavu

Put #2, , EtumerkillKokLuku

Put #3, , EtumerkillIsoKokLuku

Put #4, , PieniLiukuLuku

Put #5, , Isoliukuluku

Put #6, , MaaramittMerkkijono

Put #7, , VapaaMerkkijono

Close 1, 2, 3, 4, 5, 6, 7
Tutustu NotePad -editorilla tiedostojen sisältöön ja tutki Explorer-ohjelmalla tiedostojen tietosisällön kokoa (Properties | Size). Binääritiedostojen Size vastaa tarkalleen luvun esitystarkkuutta bitteinä/tavuina. Tiedostoissa ei ole mitään muuta kuin tiedon esittämisessä tarvittavat bitit. Tekstitiedostoissa on mukana välilyöntimerkkejä sekä ainakin "End-of-File" -merkki lopussa. Huomaa erityisesti miten "PieniLiukuLuku.txt" luvun desimaaleille on käynyt.
Esimerkki kuvatiedoston lukemisesta ja kuvan tulostamisesta näytölle

Lisää Form1-ikkunaan PictureBox -olio "Picture1"
Lisää Form1-ikkunaan kolmas komentopainike "Command3"

Lisää Module1.Bas määrittely RGBTyyppi nimiselle tieto-oliolle
Type RGBTyyppi

  R As Byte

  G As Byte

  B As Byte

End Type

Kirjoita ao. koodi Command3 -painikkeeseen tapahtumalle "Click" ja suorita koodi ("F5"  tai Run | Start)
Private Sub Command3_Click()

Rem Säädetään Picture1-olion kokoa

Picture1.ScaleMode = vbPixels

Picture1.Width = 140

Picture1.Height = 200

Rem Määritellään taulukko, joka sisältää kuvan RGB-arvot

Rem RG-tyyppi on määritelty MODULE1.BAS :ssa

ReDim kuva(1 To 140, 1 To 200) As RGBTyyppi

Rem Avataan kuva binaarimoodissa, luetaan Get-funktiolla koko

Rem tiedosto kerrallaan taulukkoon kuva()

Open "d:\minv12\mallikuva_140_200.raw" For Binary As 1

Get #1, , kuva

Close 1

Rem Tyhjennetään Picture1-olion bittikartta

Picture1.Cls

Rem Määritellään kaksi laskurimuuttujaa, isoja kokonaislukuja

Dim i As Long, j As Long

Rem Kahden For-luupin sisällä väritetään Picture1-olio piste kerrallaan

Rem Pset-funktiolla. RGB-arvot funktiolle RGB() saadaan taulukosta 
Rem kuva().

For i = 1 To 140

  For j = 1 To 200

   Picture1.PSet (i, j), RGB(kuva(i, j).R, kuva(i, j).G, kuva(i, j).B)

  Next j

Next i

Rem Kirjoitetaan tervehdys print funktiolla

Picture1.CurrentX = 10

Picture1.CurrentY = 59

Picture1.FontName = "Arial"

Picture1.FontSize = 10

Picture1.Print "Hmmm..."

End Sub
Aliohjelmat ja funktiot
Hyvä ohjelmointitapa edellyttää, etteivät yksittäiset aliohjelmat (jonkin asian toteuttavat koodikokonaisuudet) muodostu liian pitkiksi. Jonkin loogisen asian toteuttavat rutiinit voi kirjoittaa (lauserakenteisilla ohjelmointikielillä) aliohjelmiksi tai funktioiksi.

Aliohjelma

Kirjoita Module1.bas koodi-ikkunaan

Public Sub Aliohjelma(ByRef SisaanUlos As Long, ByVal Sisaan As Long)

( VB täydentää automaattisesti aliohjelman "End Sub" -rivillä

Kirjoitetaan tähän väliin aliohjelma, jolle välitetään kaksi kokonaislukua, joista toinen välittyy s.e. aliohjelma muokatessa sen sisältöä muokattu arvo välittyy takaisin ja toinen välittyy alkuperäisessä arvossaan takaisin.
Public Sub Aliohjelma(ByRef SisaanUlos As Long, ByVal Sisaan As Long)

SisaanUlos = SisaanUlos + 2

Sisaan = Sisaan + 2

End Sub
Lisää Form1-ikkunaan neljäs komentopainike "Command4" ja koodaa sen Click tapahtumaan aliohjelmakutsu

Private Sub Command4_Click()

Dim Muuttuja1 As Long, Muuttuja2 As Long

Muuttuja1 = 2

Muuttuja2 = 2

Call Aliohjelma(Muuttuja1, Muuttuja2)

MsgBox ("Arvot ovat: " & Muuttuja1 & " " & Muuttuja2)

End Sub
Funktio

Kirjoita Module1.bas koodi-ikkunaan

Public Function Summa( Byval L1 As Double, ByVal L2 As Double) as double
Summa = L1 + L2

End Function

Lisää Command4-painikkeeseen Click-tapahtumaan lisärivi loppuun
MsgBox ("Ulostulleiden summa on: " & Summa (Muuttuja1,Muuttuja2))

Ohjausrakenteet

VB:ssa on käytettävissä

if.. then..Else -rakenne 

Goto-lauseet (vältä)

Select Case -rakenne
On ... goto/Gosub -rakenne
if (totuusarvolauseke) then

  Koodi_Joka_Suoritetaan_Jos_Lauseke_Toteutui

  Else if...

end if

Aisioden toistamista varten löytyvät

For-loopit
Do While/Until (lauseke) Loop -toistorakenne

While..Wend -toistorakenne
Lisää Form1-ikkunaan komentopainike Command5 ja sen Click-tapahtumaan ao. koodi.
Private Sub Command5_Click()
Rem If Then Else -rakenne

If 1 + 1 = 2 Then

  MsgBox ("Yhteenlasku tuotti vertailtavaksi samat luvut, tosi!")

End If

If 1 + 1 = 3 Then

  MsgBox ("Tämän lauseen ei pitäisi toteutua")

  Else

  MsgBox ("Lauseke oli epätosi")

End If

Rem Goto-lause

GoTo Rivi7

  MsgBox ("Tätäkään ei koskaan näy ruudulla goto:n takia")

Rivi7: MsgBox ("Tähän tultiin Goto lauseella")

Rem Select Case Rakenne

Dim i As Long

i = Int(Rnd * 5)

Select Case i

 Case 0

 MsgBox ("Sattuma toi tähän, Rnd sai arvon 0 - 0.199999")

 Case 1

 MsgBox ("Sattuma toi tähän, Rnd sai arvon 0.2 - 0.399999")

 Case 2

 MsgBox ("Sattuma toi tähän, Rnd sai arvon 0.4 - 0.599999")

 Case 3

 MsgBox ("Sattuma toi tähän, Rnd sai arvon 0.6 - 0.799999")

 Case 4

 MsgBox ("Sattuma toi tähän, Rnd sai arvon 0.8 - 0.999999")

 Case 5

 MsgBox ("Harvinainen Sattuma toi tähän, Rnd sai 1.00000")

 End Select

Rem On  .. goto rakenne

On Error Resume Next

Dim Tavu As Byte

Rem fataali virhe ! lukualue 0.-255

Tavu = 256

Rem fataali virhe ! nollalla jako

Tavu = 100 / 0

Rem For-Next toistorakenne

A1 = Time

Dim j As Long

Rem sata miljoonaa toistoa

For i = 1 To 100000000

 j = j + 1

Next i

A2 = Time

MsgBox ("Aikaa : " & (A2 - A1) * 24 * 60 * 60 & " sek, laskut 32 bit")

Rem sata miljoonaa toistoa

Dim k As Double, l As Double

For l = 1 To 100000000

 k = k + 1

Next l

A2 = Time

MsgBox ("Aikaa : " & (A2 - A1) * 24 * 60 * 60 & "  sek, laskut 64 bit")

Rem Do-rakenne

i = 0

Do Until i = 5

MsgBox (i)

i = i + 1

Loop

MsgBox ("i:n arvo nyt " & i)

End Sub

